Dialog Control Language - Part 4 (Phần thực hành 4 về DCL)


  Part 4 - PopUp_List (Phần 4- Danh sách thả xuống)  



   Let's build a working DCL file showing us exactly how to handle popup list. (Ta sẽ xây dựng một tập tin DCL hoạt động để chỉ cho thấy chính xác cách điều khiển danh sách thả xuống )
 We will build a DCL file containing two pop_up list plus an Okay and Cancel button.   The selected items will be displayed on the screen after the user presses the Okay button. (Ta sẽ tạo một tập tin DCL chứa hai danh sách thả xuống với bộ nút Okay và Cancel. Các khoản mục được chọn sẽ hiển thị trên màn hình sau khi người sử dụng nhấn nút Okay)


  Layout thoughts:   I will place the popup_list in a row, (side by side).  Then I'll put the Okay and Cancel buttons in a row at the bottom of the dialog box.  So...I'll need something like this:
Thiết kế hộp thoại: Tôi sẽ đặt các danh sách thả trong một hàng, (nằm cạnh nhau). Sau đó tôi đặt các nút Okay và Cancel vào hàng dưới cùng của hộp thoại. Vì thế tôi sẽ cần như sau:
: column { 
  : boxed_row { 
    : popup_list { 
      // Put code for popup_list 1 here (Đặt mã cho danh sách thả xuống 1 ở đây)
    }  
    : popup_list {  
      // Put code for popup_list 2 here (Đặt mã cho danh sách thả xuống 2 ở đây) 
    }  
  }  
  : boxed_row { 
    : button { 
       // Put code for the Okay button here  (Đặt mã cho nút Okay ở đây)
    }  
    : button { 
       // Put code for the Cancel button here  (Đặt mã cho nút Cancel ở đây)
    } 
  }   
} 

Let's copy in the code for the header and all of the controls above from the "Controls" section of this tutorial.  I'll show them in red.  Notice the key names and labels had to be changed. (Ta hãy copy các mã phần tiêu đề và tất cả các nút điều khiển trên đây từ phần “Các nút điều khiển” của tài liệu tự học này. Tôi sẽ trình bày chúng bằng màu đỏ. Chú ý là các tên khoá và các nhãn đã được thay đổi. )
SAMPLE4 : dialog { 
          label = "Sample Dialog Box Routine - Part 4"; 
          : column { 
            : boxed_row { 
              : popup_list {
                key = "mylist1";
                label = "Select Item";
                fixed_width_font = true;
                width = 30;
                value = "";
              }
              : popup_list {
                key = "mylist2";
                label = "Select Item";
                fixed_width_font = true;
                width = 30;
                value = "";
              }
            }
            : boxed_row {
              : button {
                key = "accept";
                label = " Okay ";
                is_default = true;
              }
              : button {
                key = "cancel";
                label = " Cancel ";
                is_default = false;
                is_cancel = true;
              }
            }
          } 
}
Right click and copy the above. Open NotePad and paste it.  Save the file as SAMPLE4.DCL  Be sure to change the "Save as Type" drop down box to "All Files" before saving it or it will put a  ".txt" extension on the file name.  Save this file somewhere in the AutoCAD search path.


  Next we will get a copy of the AutoLisp model and revise it.  All new code is shown in red. ( Tiếp theo ta sẽ lấy một phiên bản của chương trình Autolisp mẫu và sửa đổi lại nó. Tất cả mã mới đều thể hiện bởi màu đỏ. )
(defun C:SAMPLE4()

  ;;;--- Load the dcl file
  (setq dcl_id (load_dialog "SAMPLE4.dcl"))

  ;;;--- Load the dialog definition if it is not already loaded
  (if (not (new_dialog "SAMPLE4" dcl_id))
    (progn
     (alert "The SAMPLE4.DCL file could not be loaded!")
      (exit)
    )
  )

  ;;;--- If an action event occurs, do this function
  (action_tile "accept" "(setq ddiag 2)(saveVars)(done_dialog)") ; Chú ý hàm SaveVars
  (action_tile "cancel" "(setq ddiag 1)(done_dialog)")

  ;;;--- Display the dialog box
  (start_dialog)

  ;;;--- Unload the dialog box
  (unload_dialog dcl_id)

  ;;;--- If the user pressed the Cancel button
  (if(= ddiag 1)
    (princ "\n Sample4 cancelled!")
  )

  ;;;--- If the user pressed the Okay button
  (if(= ddiag 2)
    (progn
      (princ "\n The user pressed Okay!")
    )
  )

  ;;;--- Suppress the last echo for a clean exit
  (princ)

)  

  Right click and copy the above. Open NotePad and paste it.  Save the file as SAMPLE4.LSP  Be sure to change the "Save as Type" drop down box to "All Files" before saving it or it will put a  ".txt" extension on the file name.  Save this file somewhere in the AutoCAD search path.


  Let's load the program and see what the DCL file looks like.  On the command line type this:

Command: (load "sample4") and press enter

  You should see this

C:Sample4 
Command:

  Now type Sample4 and press enter.  If everything went according to plan you should see this on your screen:

   Looks good but, there is nothing to choose.  Let's add some data to the popup list boxes.  We will need two list.  We will call the list for the first popup list box myList1 and the second myList2. (Trông thì ngon đấy nhưng chưa có gì để chọn cả. Ta sẽ thêm dữ liệu vào các hộp danh sách thả xuống. Ta cần hai danh sách. Gọi danh sách cho hộp danh sách thả xuống thứ nhất là myList1 và cho hộp thứ hai là myList2. )
  (setq myList1(list "Electrical" "Structural" "Plumbing" "Foundation"))
  (setq myList2(list "Plastic" "Steel" "Aluminum" "Concrete"))
Alrighty then, we have our list built.  All we have to do is put them in the dialog box. We will use the start_list, add_list, and end_list functions.  Start_list tells DCL which popup list box we are going to edit.  The identification is made by using the popup list box KEY.  The first popup list box has a key of "mylist1" and the second has a key of "mylist2".  So the start_list function would look like this: (Rồi, vậy là ta đã tạo xong các danh sách. Ta chỉ việc đặt chúng vào hộp thoại. Ta sẽ sử dụng các hàm start_list, add_list, và end_list. Hàm start_list dùng để xác định hộp danh sách nào sẽ được chỉnh sửa. Để chỉ định hộp, sử dụng khoá của hộp. Hộp danh sách thả xuống thứ nhất có khóa là “mylist1” và hộp thứ hai có khoá là “mylist2”. Vì thế hàm start_list sẽ như sau:
(start_list "mylist1" 3)   ; The 3 means we want to delete the old contents and start new. (Tham số 3 có nghĩa là ta sẽ xóa toàn bộ nội dung cũ và bắt đầu danh sách mới. )
  Next we use the add_list function to tell DCL which list to put in the popup list box.   We use the mapcar function to apply add_list to each member in the list.  Our list for the first popup list box is named myList1.  So... (Tiếp theo ta sử dụng hàm add_list để xác định mục nào cần đặt vào trong hộp danh sách thả xuống. Ta sử dụng hàm mapcar để áp dụng hàm ADD_list cho từng thành viên của danh sách. Danh sách ta dùng cho hộp danh sách thả xuống thứ nhất là myList1. Vì thế…)
(mapcar 'add_list myList1)
  Finally we use the end_list function to tell DCL to display the new contents because we are through editing the popup list box.  (Cuối cùng ta sử dụng hàm end_list để hiển thị nội dung mới của hộp danh sách thả xuống vì ta đã chỉnh sửa xong.)
(end_list)
To look at it all together: ( Gộp chúng lại với nhau: )
(start_list "mylist1" 3)
(mapcar 'add_list myList1)
(end_list)
(start_list "mylist2" 3)
(mapcar 'add_list myList2)
(end_list)
 

  Let's add all of this to the AutoLisp program and see what it looks like. (Thêm tất cả các mã trên vào chương trình Autolisp và ta thấy như sau:
(defun C:SAMPLE4()
  (setq myList1(list "Electrical" "Structural" "Plumbing" "Foundation")) 
  (setq myList2(list "Plastic" "Steel" "Aluminum" "Concrete"))
  ;;;--- Load the dcl file
  (setq dcl_id (load_dialog "SAMPLE4.dcl"))

  ;;;--- Load the dialog definition if it is not already loaded
  (if (not (new_dialog "SAMPLE4" dcl_id))
    (progn
     (alert "The SAMPLE4.DCL file could not be loaded!")
      (exit)
    )
  )
  (start_list "mylist1" 3)
  (mapcar 'add_list myList1)
  (end_list)
  (start_list "mylist2" 3)
  (mapcar 'add_list myList2)
  (end_list)
  ;;;--- If an action event occurs, do this function
  (action_tile "accept" "(setq ddiag 2)(saveVars)(done_dialog)")
  (action_tile "cancel" "(setq ddiag 1)(done_dialog)")

  ;;;--- Display the dialog box
  (start_dialog)

  ;;;--- Unload the dialog box
  (unload_dialog dcl_id)

  ;;;--- If the user pressed the Cancel button
  (if(= ddiag 1)
    (princ "\n Sample4 cancelled!")
  )

  ;;;--- If the user pressed the Okay button
  (if(= ddiag 2)
    (progn
      (princ "\n The user pressed Okay!")
    )
  )

  ;;;--- Suppress the last echo for a clean exit
  (princ)

)  

  Notice the location of the red lines.  We create the list before loading the dialog box.  We add the list to the dialog box after it is loaded with new_dialog and before the action_tile statements.  This is the order you should use. (Chú ý tới vị trí của các dòng màu đỏ. Ta tạo danh sách trước khi tải hộp thoại. Thêm danh sách vào hộp thoại sau khi nó đã được tải bởi hàm new_dialog và trước hàm thông báo phần tử hành động action_tile. Đó là trật tự mà bạn nên sử dụng.)
   Looking good so far.  We need to add the SaveVars function to save the selected items from the popup list boxes when the Okay button is pressed.  Look at the blue text in the Sample4.lsp program above.(Tiếp tục nữa nhé. Ta cần phải thêm hàm SaveVars để lưu lại các khoản mục đã được chọn từ hộp danh sách thả xuống khi nút Okay bị nhấn. Hãy xem dòng chữ màu xanh trong chương trình Autolisp Sample4.lsp ở trên. )
  Let's steal the saveVars routine from the popup list box control on the "Saving data from the dialog box" page of this tutorial and modify it.  I'll show the modifications in red. (Ta sẽ chôm vòng lưu biến SaveVars từ mục điều khiển hộp danh sách thả xuống  trong phần “Lưu dữ liệu từ hộp thoại” của tài liệu này và sửa đổi lại. Tôi sẽ thể hiện các phần sửa đổi bằng màu đỏ.)
(defun saveVars()
  ;;;--- Get the selected item from the first list
  (setq sStr1(get_tile "mylist1")) 

  ;;;--- Make sure something was selected... 
  (if(= sStr1 "")> 
    (setq myItem1 nil) 
    (setq myItem1 (nth (atoi sStr1) myList1)) 
  ) 

) 

  Wow! That was easy.  Wait a minute, we have two pop_up list boxes.  We will have to create a function out of this or simply copy this and do it twice.  For now, let's just do it twice. (Ồ, dễ quá nhỉ. Khoan đã đừng tưởng bở. Ta có hai hộp danh sách thả xuống cơ mà. Ta sẽ phải tạo một hàm cho cả hai hộp danh sách này hay đơn giản hơn ta copy điều này hai lần. Ở đây ta sẽ thực hiện hai lần copy.)

Our program would now look like this: (Chương trình của ta trông giống như sau: ) 

(defun saveVars()
  ;;;--- Get the selected item from the first list
  (setq sStr1(get_tile "mylist1")) ;;; Hàm (get_tile “mylist1”) trả về số thứ tự của khoản mục được chọn .
  ;;;--- Make sure something was selected... 
  (if(= sStr1 "")>    ;;; === Symbol “>” is excess and needs to obmit ===;;;
    (setq myItem1 "Nothing") 
    (setq myItem1 (nth (atoi sStr1) myList1)) 
  ) 

  ;;;--- Get the selected item from the second list
  (setq sStr2(get_tile "mylist2")) 

  ;;;--- Make sure something was selected... 
  (if(= sStr2 "")>   ;;; === Symbol “>” is excess and needs to obmit ===;;;
    (setq myItem2 "Nothing") 
    (setq myItem2 (nth (atoi sStr2) myList2)) 
  ) 
) 

(defun C:SAMPLE4()
  (setq myList1(list "Electrical" "Structural" "Plumbing" "Foundation")) 
    (setq myList2(list "Plastic" "Steel" "Aluminum" "Concrete"))
  ;;;--- Load the dcl file
  (setq dcl_id (load_dialog "SAMPLE4.dcl"))

  ;;;--- Load the dialog definition if it is not already loaded
  (if (not (new_dialog "SAMPLE4" dcl_id))
    (progn
     (alert "The SAMPLE4.DCL file could not be loaded!")
      (exit)
    )
  )
  (start_list "mylist1" 4)
  (mapcar 'add_list myList1)
  (end_list)
  (start_list "mylist2" 4)
  (mapcar 'add_list myList2)
  (end_list)
  ;;;--- If an action event occurs, do this function
  (action_tile "accept" "(setq ddiag 2)(saveVars)(done_dialog)")
  (action_tile "cancel" "(setq ddiag 1)(done_dialog)")

  ;;;--- Display the dialog box
  (start_dialog)

  ;;;--- Unload the dialog box
  (unload_dialog dcl_id)

  ;;;--- If the user pressed the Cancel button
  (if(= ddiag 1)
    (princ "\n Sample4 cancelled!")
  )

  ;;;--- If the user pressed the Okay button
  (if(= ddiag 2)
    (progn
      (princ "\n The user pressed Okay!")
    )
  )

  ;;;--- Suppress the last echo for a clean exit
  (princ)

)  



  Last item.  We need to replace the line in the program: (princ "\n The user pressed Okay!")  with something to modify and display the selected items.  Let's do something simple.  We will tell the user what was selected out of each popup list..  (Cuối cùng, ta cần thay thế dòng (princ “\n The user pressed Okay!” ) bằng cái gì đó để sửa đổi và hiển thị các khoản mục được chọn. Hãy làm điều đơn giản này. Báo cho người sử dụng biết cái gì đã được chọn ra từ danh sách thả xuống)
  ;;;--- If the user pressed the Okay button
  (if(= ddiag 2)
    (progn
      ;;;--- Inform the user of his selection from the first list
      (princ (strcat "\n You chose " myItem1 " from the first popup list box."))
      ;;;--- Inform the user of his selections from the second list 
      (princ (strcat "\n You chose " myItem2 " from the second popup list box."))

    )
  )


Add the above to the file, save it and test it out.   Everything working okay? (Bổ sung những điều này vào tập tin, lưu lại và chạy thử nó. Mọt thứ làm việc tốt cả chứ ?)

   When you get your program tested and everything is working, move the blue line above, [ (defun C:SAMPLE4() ] all the way to the top of the file.  This will make all of your variables local and will reset them all to nil when the program ends.

  That's it.  We're done.

